Limite d'un fonction de deux variables, une introduction

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en janvier 2006. L'objectif de ce document Maple est de présenter de manière intuitive la limite d'une fonction de deux variables. Dans ce document, on abordera la notion de limite de manières numérique et graphique. Quelques exemples illustreront graphiquement quelques chemins d'Chemins dans l'évaluation de la limite en un point d'une fonction.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,display,spacecurve,setoptions3d);
  setoptions3d(size=[300,300],labels=[x,y,z],axesfont=[TIMES,ROMAN,8],
  labelfont=[TIMES,ROMAN,8],titlefont=[TIMES,ITALIC,14],axes=normal);
```

$$[display, spacecurve, setoptions3d]$$

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimamt les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de composition
  étendue
  Typesetting:-Settings(striptrailing=true);
```

$$extended$$

$$false$$

```
> interface(displayprecision=8):
```

Deux exemples de fonctions

Soit la fonction définie par $f(x, y) = \dfrac{\sin(x^2 + y^2)}{x^2 + y^2}$.

```
> f:=(x,y)->sin(x^2+y^2)/(x^2+y^2);
```

$$f := (x, y) \mapsto \frac{\sin(y^2 + x^2)}{y^2 + x^2}$$

Construisons un tableau de valeurs pour observer le comportement de la fonction f au voisinage de $(0, 0)$.
(Après avoir exécuté entièrement la feuille de travail, modifiez la taille des résultats des cellules du tableau à la taille 9 pour ajuster correctement les nombres dans les cellules)<

Table 1: Fonction f au voisinage de (0,0)

f(x, y)	-1	-0.5	-0.2	-0.1	0	0.1	0.2	0.5	1
-1	0.45464871	0.7591877	0.82923483	0.83844737	0.84147098	0.83844737	0.82923483	0.7591877	0.45464871
-0.5	0.7591877	0.95885108	0.98604216	0.98877135	0.98961584	0.98877135	0.98604216	0.95885108	0.7591877
-0.2	0.82923483	0.98604216	0.99893367	0.99958339	0.99973335	0.99958339	0.99893367	0.98604216	0.82923483
-0.1	0.83844737	0.98877135	0.99958339	0.99993333	0.99998333	0.99993333	0.99958339	0.98877135	0.83844737
0	0.84147098	0.98961584	0.99973335	0.99998333		0.99998333	0.99973335	0.98961584	0.84147098
0.1	0.83844737	0.98877135	0.99958339	0.99993333	0.99998333	0.99993333	0.99958339	0.98877135	0.83844737
0.2	0.82923483	0.98604216	0.99893367	0.99958339	0.99973335	0.99958339	0.99893367	0.98604216	0.82923483
0.5	0.7591877	0.95885108	0.98604216	0.98877135	0.98961584	0.98877135	0.98604216	0.95885108	0.7591877
1	0.45464871	0.7591877	0.82923483	0.83844737	0.84147098	0.83844737	0.82923483	0.7591877	0.45464871

Il semble que $f(x, y) \to 1$ lorsque $(x, y) \to (0,0)$.

Avec le même voisinage de $(0, 0)$, observons les images de la fonction g définie par $g(x, y) = \dfrac{x^2 - y^2}{x^2 + y^2}$.

```
> g:=(x,y)->(x^2-y^2)/(x^2+y^2);
```

$$g := (x, y) \mapsto \frac{x^2 - y^2}{y^2 + x^2}$$

Table 2: Fonction g au voisinage de (0,0)

g(x, y)	-1	-0.5	-0.2	-0.1	0	0.1	0.2	0.5	1
-1	0.	0.6	0.92307692	0.98019802	1.	0.98019802	0.92307692	0.6	0.
-0.5	−0.6	0.	0.72413793	0.92307692	1.	0.92307692	0.72413793	0.	−0.6
-0.2	−0.92307692	−0.72413793	0.98019802	0.6	1.	0.6	0.	−0.72413793	−0.92307692
-0.1	−0.98019802	−0.92307692	−0.6	0.	1.	0.	−0.6	−0.92307692	−0.98019802
0	−1.	−1.	−1.	−1.		−1.	−1.	−1.	−1.
0.1	−0.98019802	−0.92307692	−0.6	0.	1.	0.	−0.6	−0.92307692	−0.98019802
0.2	−0.92307692	−0.72413793	0.	0.6	1.	0.6	0.	−0.72413793	−0.92307692
0.5	−0.6	0.	0.72413793	0.92307692	1.	0.92307692	0.72413793	0.	−0.6
1	0.	0.6	0.92307692	0.98019802	1.	0.98019802	0.92307692	0.6	0.

Dans le cas de la fonction g par contre, lorsque (x, y) se rapproche de $(0, 0)$, les images de g ne semble pas clairement se rapprocher d'un nombre L

Graphiques des fonctions f et g

Observons le comportement des fonctions f et g à l'aide de leur tracé respectif au voisinage de (0,0).

Traçons d'abord la surface définie par la fonction f sur le pavé $[-1, 1] \times [-1, 1]$.

```
> Graphe_f:=plot3d([x,y,f(x,y)],x=-1..1,y=-1..1):
  display(Graphe_f);
```



Zoomons légèrement vers l'avant.

```
> Graphe_f:=plot3d([x,y,f(x,y)],x=-0.25..0.25,y=-0.25..0.25):
  display(Graphe_f);
```



Zoomons encore un peu plus vers l'avant.

```
> Graphe_f:=plot3d([x,y,f(x,y)],x=-0.1..0.1,y=-0.1..0.1):
  display(Graphe_f,axes=boxed);
```



Avec une approche numérique et graphique de la fonction f, il semble que si (x, y) se rapproche de $(0, 0)$, $f(x, y)$ se rapproche de 1.

Faisons de même pour la fonction g.

```
> Graphe_g:=plot3d([x,y,g(x,y)],x=-1..1,y=-1..1):
  display(Graphe_g,axes=boxed,orientation=[60,65]);
```



En cliquant sur la figure précédente tout en maintenant enfoncé le bouton gauche de la souris, on peut orienter manuellement la surface dans l'espace pour mieux en saisir les caractéristiques. La fissure que l'on observe n'est pas une caractéristique de cette surface. En effet, le domaine de la fonction g est $\{(x,y) | (x,y) \neq (0,0)\}$. Donc, la surface aurait dû être davantage rapproché de la cote en donnant même l'impression qu'elle touche à cet axe de coordonnées plutôt que de faire apparaître une telle fissure. Cela est dû à une limitation de la macro-commande `plot3d` que l'on rencontre dans des situtations où le calcul de la fonction comporte des quotients par exemple. Dans certains cas, on peut améliorer le tracé de la surface en augmentant le nombre de points dans les calculs à l'aide de l'option `grid`.

Zoomons légèrement vers l'avant pour observer de plus près la surface.

```
> Graphe_g:=plot3d([x,y,g(x,y)],x=-0.1..0.1,y=-0.1..0.1):
  display(Graphe_g,axes=boxed,orientation=[60,65]);
```



Nous observons alors un comportement de la fonction g très différent de la fonction f au voisinage de $(0,0)$.

Remarque: Il est possible d'améliorer le tracé de cette surface. Obtenons le tracé de g (au voisinage de $(0,0)$) par morceaux afin que `plot3d` calcul plus correctement les images de g dans chaque morceau de ce voisinage).

```
> Graphe_g1:=plot3d([x,y,g(x,y)],x=-1..1,y=-1..1,grid=[120,240],thickness=0):
  Graphe_g2:=plot3d([x,y,g(x,y)],x=-1..1,y=-1..1,grid=[120,240],thickness=0):
  Graphe_g3:=plot3d([x,y,g(x,y)],x=-1..1,y=-1..1,grid=[120,240],thickness=0):
  Graphe_g4:=plot3d([x,y,g(x,y)],x=-1..1,y=-1..1,grid=[120,240],thickness=0):
  Surface:=display(Graphe_g||(1..4),style=patchnogrid,transparency=0.0,
```

```
orientation=[60,65]):
> display(Surface,axes=boxed);
```



Calculs des limites des fonctions f et g en (0,0)

Calculons la limite de la fonction f lorsque (x, y) se rChemin de $(0, 0)$.

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

$$Limit\left(\frac{\sin(x^2+y^2)}{x^2+y^2}, \{x=0, y=0\}\right) = 1$$

Nous avions déjà observé de deux manières différentes le comportement de la fonction f au voisinage de $(0, 0)$ et qu'il semble que $f(x, y)$ se rapproche de 1 lorsque (x, y) se rapproche de $(0, 0)$. Maple confirme que la limite est égale à 1.

Calculons la limite de la fonction g lorsque (x, y) se rapproche de $(0, 0)$.

```
> Limit(g(x,y),{x=0,y=0})=limit(g(x,y),{x=0,y=0});
```

$$Limit\left(\frac{x^2-y^2}{x^2+y^2}, \{x=0, y=0\}\right) = -1\,..1$$

C'est de cette façon dont Maple s'exprime pour signifier qu'il y a différentes limites de la fonction g selon les chemins utilisés dans le plan pour se rapprocher de $(0, 0)$. La limite demandée n'existe donc pas.

Limites avec différentes chemins dans le domaine de la fonction

```
> Plan:=plot3d([x,y,0],x=-1..1,y=-1..1,style=wireframe,color="Olive",grid=[10,
  10],transparency=0.8):
```

Approchons (0,0) avec $y = 0$.

```
> Chemin_1:=spacecurve([x,0,0],x=-1..1,color="Burgundy",thickness=2):
  Image_Chemin_1:=spacecurve([x,0,g(x,0)],x=-1..1,thickness=2,color=
  "DeepBlue"):
> display([Surface,Chemin_1,Image_Chemin_1],axes=boxed);
```




```
> display([Plan,Chemin_1,Image_Chemin_1]);
```




```
> Limit(g(x,0),x=0)=limit(g(x,0),x=0);
```

$$\lim_{x \to 0} 1 = 1$$

Approchons (0,0) avec $x = 0$.

```
> Chemin_2:=spacecurve([0,y,0],y=-1..1,color="Burgundy",thickness=2):
  Image_Chemin_2:=spacecurve([0,y,g(0,y)],y=-1..1,thickness=2,
  color="DeepBlue"):
```

```
> display([Surface,Chemin_2,Image_Chemin_2],axes=boxed);
```




```
> display([Plan,Chemin_2,Image_Chemin_2]);
```



```
> Limit(g(0,y),y=0)=limit(g(0,y),y=0);
```

$$\lim_{y \to 0} (-1) = -1$$

Approchons (0,0) avec $y = x^2$.

```
> Chemin_3:=spacecurve([x,x^2,0],x=-1..1,color="Burgundy",thickness=2):
  Image_Chemin_3:=spacecurve([x,x^2,g(x,x^2)],x=-1..1,thickness=2,
  color="DeepBlue"):
```

```
> display([Surface,Chemin_3,Image_Chemin_3],axes=boxed);
```



```
> display([Plan,Chemin_3,Image_Chemin_3]);
```



```
> Limit(g(x,x^2),x=0)=limit(g(x,x^2),x=0);
```

$$\lim_{x \to 0} \frac{-x^4 + x^2}{x^4 + x^2} = 1$$

Approchons $(0, 0)$ avec $x = y^2$.

```
> Chemin_4:=spacecurve([y^2,y,0],y=-1..1,color="Burgundy",thickness=2):
```

```
Image_Chemin_4:=spacecurve([y^2,y,g(y^2,y)],y=-1..1,thickness=2,
color="DeepBlue"):
```

```
> display([Surface,Chemin_4,Image_Chemin_4],axes=boxed);
```



```
> display([Plan,Chemin_4,Image_Chemin_4],orientation=[45,70]);
```



```
> Limit(g(y^2,y),y=0)=limit(g(y^2,y),y=0);
```

$$\lim_{y \to 0} \frac{y^4 - y^2}{y^4 + y^2} = -1$$

Quelle est la conclusion qui s'impose quant à la limite?

Autre exemple

Soit la fonction f définie par $f(x, y) = \frac{x y^2}{x^2 + y^4}$.

```
> f:=(x,y)->x*y^2/(x^2+y^4);
```

$$f := (x, y) \mapsto \frac{x \cdot y^2}{x^2 + y^4}$$

Obtenons le graphique de la surface d'équation $z = \frac{x y^2}{x^2 + y^4}$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-1..1,y=-1..1,grid=[120,120]):
display(Surface,style=patchnogrid,orientation=[60,75]);
```



Remarque: Encore ici, puisque le domaine de cette fonction est $\mathbb{R}^2 \setminus \{(0,0)\}$, le tracé de la surface devrait davantage être rapproché de la cote en donnant l'impression qu'elle touche à cet axe. Nous allons nous contenter de cette représentation telle quelle.

Calculons la limite lorsque (x,y) se rapproche de $(0,0)$.

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

$$\mathit{Limit}\left(\frac{x y^2}{y^4+x^2}, \{x=0, y=0\}\right) = -\frac{1}{2} .. \frac{1}{2}$$

Approchons $(0,0)$ avec le chemin $y = 2x$.

```
> Limit(f(x,2*x),x=0)=limit(f(x,2*x),x=0);
```

$$\lim_{x \to 0} \frac{4x^3}{16x^4+x^2} = 0$$

Illustrons graphiquement cette Chemin.

```
> Plan:=plot3d([x,y,0],x=-1..1,y=-1..1,style=wireframe,color="Olive",grid=[10,
  10],transparency=0.8):
  Chemin_1:=spacecurve([x,2*x,0],x=-1/2..1/2,color="Burgundy",thickness=3):
  Image_Chemin_1:=spacecurve([x,2*x,f(x,2*x)],x=-1/2..1/2,thickness=2,
  color="DeepBlue"):
```

```
> display([Surface,Plan,Chemin_1,Image_Chemin_1],
          axes=boxed,lightmodel=light4,
          style=patchnogrid,orientation=[30,60,0]);
```



Affichons de nouveau cette approche mais en omettant le tracé de la surface dans le graphique.

```
> display([Plan,Chemin_1,Image_Chemin_1],orientation=[30,60,0]);
```



Voyons si le chemin d'approche avec n'importe quelle droite de pente réelle dans le domaine passant par l'origine d'équations $y = mx$ donnera la même limite, soit 0. Considérons les pentes suivantes:

```
> Valeurs_de_pente:=[-3,-2,-1,1,2,3];
```

$$Valeurs\_de\_pente := [-3, -2, -1, 1, 2, 3]$$

```
> k:=0:
  for m in Valeurs_de_pente do
    k:=k+1;
    Chemin||k:=spacecurve([x,m*x,0],x=-abs(1/m)..abs(1/m),color="Burgundy",
  thickness=2):
    Image_Chemin||k:=spacecurve([x,m*x,f(x,m*x)],x=-abs(1/m)..abs(1/m),
                         thickness=2,color="DeepBlue")
  end do:
> display([Surface,seq(Chemin||k,k=1..6),seq(Image_Chemin||k,k=1..6)],style=
  patchnogrid);
```



Ometttons la surface dans le graphique pour mieux voir.

```
> display([Plan,seq(Chemin||k,k=1..6),seq(Image_Chemin||k,k=1..6)
  ],orientation=[30,60,0]);
```



Il semble effectivement que cela donne toujours 0 quelque soit le chemin d'approche linéaire de pente réelle passant par l'origine.

Calculons alors avec Maple la limite de la fonction f en $(0, 0)$ avec toute Chemin de la forme $y = m x$.

```
> Limit(f(x,m*x),x=0)=limit(f(x,m*x),x=0);
```

$$\lim_{x \to 0} \frac{9 x^3}{81 x^4 + x^2} = 0$$

Avant de conclure que la limite est 0 (même s'il y a une infinité d'approches qui donnent 0 pour limite), il faut noter que nous n'avons pas considérer toutes les trajectoires d'approches possibles. Nous avons considérer une infinité d'approches linéaires mais il y a aussi d'autres chemins possibles autres que linéaires.

Approche avec la parabole d'équation $x = y^2$.

```
> Chemin_p:=spacecurve([y^2,y,0],y=-1..1,color="Burgundy",thickness=3):
  Image_Chemin_p:=spacecurve([y^2,y,f(y^2,y)],y=-1..1,thickness=3,
  color="DeepBlue"):
```

```
> display([Surface,Chemin_p,Image_Chemin_p],style=patchnogrid);
```



```
> display([Plan,Chemin_p,Image_Chemin_p]);
```



Il semble que lorsque (x, y) se rapproche de $(0, 0)$ avec le chemin $y = x^2$, f(x,y) s'approche de $\frac{1}{2}$ et non plus de 0.

Calcul de la limite en approche de $(0, 0)$ selon la parabole d'équation $x = y^2$.

```
> Limit(f(y^2,y),y=0)=limit(f(y^2,y),y=0);
```

$$\lim_{y \to 0} \frac{1}{2} = \frac{1}{2}$$

Ce qui nous permet de conclure immédiatement que la limite demandée n'existe pas.

Un dernier calcul de limite, juste pour voir!

Calcul de la limite selon la parabole d'équation $x = -y^2$

```
> Limit(f(-y^2,y),y=0)=limit(f(-y^2,y),y=0);
```

$$\lim_{y \to 0} -\frac{1}{2} = -\frac{1}{2}$$

Illustrons cette limite.

```
> Chemin_p:=spacecurve([-y^2,y,0],y=-1..1,color="Burgundy",thickness=2):
  Image_Chemin_p:=spacecurve([-y^2,y,f(-y^2,y)],y=-1..1,thickness=2,
  color="DeepBlue"):
```

```
> display([Surface,Chemin_p,Image_Chemin_p],
          lightmodel=none,style=patchnogrid);
```



```
> display([Plan,Chemin_p,Image_Chemin_p]);
```



Un dernier exemple

Soit la fonction f définie par $f(x,y) = \arctan\left(\frac{y}{x}\right)$.

La fonction rationnelle $\frac{y}{x}$ étant continue sur son domaine $\mathbb{R}^2 \setminus \{(x,y) | x = 0\}$ et la fonction arctan étant elle continue sur $\mathbb{R}$, la fonction composée $\arctan\left(\frac{y}{x}\right)$ est alors continue sur $\{(x,y) | x \in \mathbb{R}^*, y \in \mathbb{R}\}$.

Reproduisons le mêmes étapes que l'exemple précédent.

```
> f:=(x,y)->arctan(y/x);
```

$$f := (x,y) \mapsto \arctan\left(\frac{y}{x}\right)$$

```
> Plan:=plot3d([x,y,0],x=-2..2,y=-2..2,style=wireframe,color="Olive",grid=[10,
  10],transparency=0.8):
  Graphe_f1:=plot3d([x,y,f(x,y)],x=-2..-0.00000001,y=-2..2,grid=[60,60]):
  Graphe_f2:=plot3d([x,y,f(x,y)],x=0.00000001..2,y=-2..2,grid=[60,60]):
  Surface:=display(Graphe_f1,Graphe_f2,
      axes=boxed,lightmodel=none,
      style=patchnogrid,orientation=[70,55]):
  display(Surface);
```



```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

$$\mathit{Limit}\left(\arctan\left(\frac{y}{x}\right), \{x=0, y=0\}\right) = \mathit{undefined}$$

Approchons $(0, 0)$ avec $y = x$.

```
> Chemin_1:=spacecurve([x,x,0],x=-2..2,color="Burgundy",thickness=2):
  Image_Chemin_1:=spacecurve([x,x,f(x,x)],x=-2..2,thickness=2,
  color="DeepBlue"):
```

```
> display(Surface,Plan,Chemin_1,Image_Chemin_1,orientation=[80,80]);
```



```
> display(Plan,Chemin_1,Image_Chemin_1,orientation=[80,80]);
```




```
> Limit(f(x,x),x=0)=limit(f(x,x),x=0);
```

$$\lim_{x \to 0} \frac{\pi}{4} = \frac{\pi}{4}$$

Finalement, approchons $(0, 0)$ avec $y = -x$.

```
> Chemin_2:=spacecurve([x,-x,0],x=-2..2,color="Burgundy",thickness=2):
  Image_Chemin_2:=spacecurve([x,-x,f(x,-x)],x=-2..2,thickness=2,
  color="DeepBlue"):
```

```
> display(Surface,Plan,Chemin_2,Image_Chemin_2,orientation=[55,75]);
```



```
> display(Plan,Chemin_2,Image_Chemin_2,orientation=[55,75]);
```




```
> Limit(f(x,-x),x=0)=limit(f(x,-x),x=0);
```

$$\lim_{x \to 0} -\frac{\pi}{4} = -\frac{\pi}{4}$$

Ce qui nous permet de conclure clairement que la limite demandée n'existe pas.

Une dernière Chemin, juste pour voir! Approchons-nous de $(0, 0)$ avec $y = x^2$.

```
> Chemin_3:=spacecurve([x,x^2,0],x=-sqrt(2)..sqrt(2),color="Burgundy",
  thickness=2):
  Image_Chemin_3:=spacecurve([x,x^2,f(x,x^2)],x=-sqrt(2)..sqrt(2),thickness=2,
  color="DeepBlue"):
> display(Surface,Plan,Chemin_3,Image_Chemin_3,orientation=[55,75]);
```




```
> display(Plan,Chemin_3,Image_Chemin_3,orientation=[75,75]);
```



Le graphique montre que la limite avec le chemin $y = x^2$ est 0.

```
> Limit(f(x,x^2),x=0)=limit(f(x,x^2),x=0);
```

$$\lim_{x \to 0} \arctan(x) = 0$$